x,clear,dambræt<
a,_n<
_b_e_g_i_n _c_o_m_m_e_n_t damspillende program;
   _i_n_t_e_g_e_r stacktop,ran,stp,oldtop,stinc,bufant,byte,maxniv,niveau,state,
   link,point,mi,t,i,ii,q,k,ant,alt,j,db,højde,bogst,goodnes,looki;
   _i_n_t_e_g_e_r _a_r_r_a_y b[1:32,1:6];
   _b_o_o_l_e_a_n newmove,quin,color,m,v,dm,di,slut,valg,look;
   _b_o_o_l_e_a_n _a_r_r_a_y buf,draw,a[0:10],stack[0:1000],hb[0:10,1:40];

   _c_o_m_m_e_n_t
b[1:32,1:6]

h  g  f e  d  c b  a

32   31   30   29          1
   28   27   26   25       2
24   23   22   21          3
   20   19   18   17       4
16   15   14   13          5
   12   11   10    9       6
 8    7    6    5          7
    4    3    2    1       8

 h    h    h    h
    h    h    h    h
 h    h    h    h
    .    .    .    .
 .    .    .    .
    s    s    s    s
 s    s    s    s
    s    s    s    s
;

   _p_r_o_c_e_d_u_r_e tostack;
   _b_e_g_i_n _i_n_t_e_g_e_r i;
      buf[bufant]:=newmove;
      _f_o_r i:=1 _s_t_e_p 1 _u_n_t_i_l bufant _d_o
      stack[stp+i-1]:=buf[i];
      stack[stp+bufant]:=deadpos;
      stack[stp+bufant+1]:=buf[0];

      stack[stp+bufant+2]:=_b_o_o_l_e_a_n(bufant+3);
      stinc:=stinc+bufant+3;
      stp:=stp+bufant+3;
      alt:=alt+1
   _e_n_d tostack;

   _p_r_o_c_e_d_u_r_e COLOR(color);
   _v_a_l_u_e color; _b_o_o_l_e_a_n color;
   writetext(_i_f color _t_h_e_n |<<hvid|> _e_l_s_e |<<sort|>);

   _i_n_t_e_g_e_r _p_r_o_c_e_d_u_r_e look forward(color);
   _v_a_l_u_e color; _b_o_o_l_e_a_n color;
   _b_e_g_i_n _i_n_t_e_g_e_r i,ii,k,t,gieri;
      _b_o_o_l_e_a_n m,v;
      _i_f _f_a_l_s_e _t_h_e_n _b_e_g_i_n writecr; writetext(|<<look forward |>); COLOR(color); _e_n_d;
      bufant:=state:=1; stp:=stacktop+2;
      link:=_i_f color _t_h_e_n 3 _e_l_s_e 5;
      alt:=stinc:=0;niveau:=niveau+1;

      _f_o_r i:=1 _s_t_e_p 1 _u_n_t_i_l 32 _d_o
      _b_e_g_i_n looki:=i;
         gieri:=gier(look);
         _i_f _f_a_l_s_e _t_h_e_n _b_e_g_i_n writecr; writetext(|<<i, gieri: |>); write(|<ddd|>,i,gieri); _e_n_d;
         _i_f gieri=2 _t_h_e_n
         _b_e_g_i_n _c_o_m_m_e_n_t felt optaget af egen farve;
            quin:=_b_o_o_l_e_a_nb[i,1]_s_h_i_f_t2;
            newmove:=buf[1]:=_4_00∨(_b_o_o_l_e_a_ni)∨(quin_s_h_i_f_t8);
            maxniv:=bufant:=1; buf[0]:=_f_a_l_s_e; byte:=10;
            move(i,_f_a_l_s_e,-1,1);
            b[i,1]:=_i_n_t_e_g_e_r(quin_s_h_i_f_t-2);
         _e_n_d gennemgang af trækmuligheder;
      _e_n_d_f_o_r;

      _i_f state=1 _t_h_e_n
      _b_e_g_i_n _c_o_m_m_e_n_t spillet tabt paa dette niveau;
         newmove:=_f_a_l_s_e;
         buf[0]:=_b_o_o_l_e_a_n(-100);
         tostack
      _e_n_d;

      stack[stacktop]:=_b_o_o_l_e_a_nniveau;
      stack[stacktop+1]:=_b_o_o_l_e_a_nstinc;
      oldtop:=stacktop;
      stacktop:=stacktop+stinc+4;
      stack[stacktop-2]:=_f_a_l_s_e;
      stack[stacktop-1]:=_b_o_o_l_e_a_n(oldtop-1);
      stack[oldtop-2]:=_b_o_o_l_e_a_n(stacktop-2);
      lookforward:=stacktop-2-(_i_n_t_e_g_e_rstack[stacktop-3])
   _e_n_d lookforward ;

   _c_o_r_e_c_o_d_e look,looki,b,color;
   3,46
   2,44
   2,48
   2,46
   arna2
   hhreLZ
   xr,mlnre2
   xr,arc42
   ara3,il
   pi(c17)
   hvnre1NOA
   arn5 12D
   ara4IOA
   hhnreLOC
   hhnreNOC
   e:arnc42,arc42
   e1:arc42,hrs1
   e2:6
   _e;

   _p_r_o_c_e_d_u_r_e  inmove(i,j);
   _v_a_l_u_e  i,j; _i_n_t_e_g_e_r  i,j;
   _c_o_m_m_e_n_t brik flyttes fra felt nr.j til felt nr.i;
   _b_e_g_i_n  _b_o_o_l_e_a_n bi1;
      bi1:=_b_o_o_l_e_a_nb[i,1];
      _i_f byte>30 _t_h_e_n

      _b_e_g_i_n  byte:=0; buf[bufant]:=newmove;
         bufant:=bufant+1; newmove:=_f_a_l_s_e
      _e_n_d;
      newmove:=newmove∨ (( _b_o_o_l_e_a_n i ∨( bi1 _s_h_i_f_t 10)) _s_h_i_f_t byte );
      byte:=byte+10;
      b[i,1]:=b[j,1]; b[j,1]:=0;
      _i_f bi1 _t_h_e_n
      point:=(_i_f bi1 _s_h_i_f_t 2 _t_h_e_n 10 _e_l_s_e 4)  _e_l_s_e point:=0;
      point:=point+ b[i,2] -  b[j,2]+(_i_f(i>28∨i<5)∧(-,buf[1]_s_h_i_f_t-8)_t_h_e_n 10 _e_l_s_e0);
      buf[0]:= _b_o_o_l_e_a_n ( _i_n_t_e_g_e_r buf[0]+point)
   _e_n_d procedure inmove;

   _p_r_o_c_e_d_u_r_e reestab(i);
   _v_a_l_u_e i;_i_n_t_e_g_e_r i;
   _c_o_m_m_e_n_t brik flyttes i felter tilbage;
   _b_e_g_i_n _b_o_o_l_e_a_n c,old; _i_n_t_e_g_e_r i1,i2,j;
      _f_o_r j:=1 _s_t_e_p 1 _u_n_t_i_l i _d_o
      _b_e_g_i_n c:= (newmove _s_h_i_f_t (10-byte))∧(_3_00_1_0m);
         i1:= _i_n_t_e_g_e_r(c∧_3_30_7m);
         old:= (c∧(_3_00_3m_70)) _s_h_i_f_t -10;

         b[i1,1]:= _i_n_t_e_g_e_r old;
         newmove:=newmove ∧(( _3_0m_1_00) _s_h_i_f_t ( byte -  10));
         byte :=byte - 10;
         _i_f byte= 0 _t_h_e_n
         _b_e_g_i_n byte := 40;
            bufant := bufant - 1;
            newmove := _b_o_o_l_e_a_n buf [ bufant]
         _e_n_d;
         i2 := _i_n_t_e_g_e_r(( newmove _s_h_i_f_t ( 10 - byte)) ∧_3_30_7m);
         _i_f old _t_h_e_n
         point := ( _i_f old _s_h_i_f_t 2 _t_h_e_n 10 _e_l_s_e 4) _e_l_s_e point := 0;
         point := point + b[ i1,2] - b[ i2,2]+(_i_f(i1>28∨i1<5)∧(-,buf[1]_s_h_i_f_t -8) _t_h_e_n10_e_l_s_e0);
         buf[0] := _b_o_o_l_e_a_n ( _i_n_t_e_g_e_r buf[0] - point )
      _e_n_d for loop
   _e_n_d procedure reestab;

   _p_r_o_c_e_d_u_r_e move(place,take,bad,niveau);
   _v_a_l_u_eplace,take,bad,niveau;
   _b_o_o_l_e_a_ntake;_i_n_t_e_g_e_rplace,bad,niveau;

   _b_e_g_i_n _i_n_t_e_g_e_r j,sq,sqo,gieri;
      _i_f _f_a_l_s_e _t_h_e_n _b_e_g_i_n writecr; writetext(|<<move|>); writetext(_i_f quin _t_h_e_n |<< dam|> _e_l_s_e |<< alm|>); _e_n_d;
      _i_f -,quin _t_h_e_n
      _b_e_g_i_n _c_o_m_m_e_n_t almindelig brik;
         _f_o_r j:=link,link+1 _d_o
         _b_e_g_i_n sq:=looki:=b[place,j];
            gieri:=gier(look);
            _i_f _f_a_l_s_e _t_h_e_n _b_e_g_i_n writecr; writetext(|<<  looki, gieri: |>); write(|<ddd|>,looki,gieri); _e_n_d;
            _c_a_s_e gieri _o_f
            _b_e_g_i_n

               _c_o_m_m_e_n_t frit felt;
               _i_f -,take ∧ state_<2 _t_h_e_n
               _b_e_g_i_n state:=2;
                  inmove(sq,place); tostack;
                  reestab(1)
               _e_n_d case1;

               _c_o_m_m_e_n_t felt opt. af egen farve;;

               _c_o_m_m_e_n_t felt opt. af modstander;
               _i_f state_<3 _t_h_e_n
               _b_e_g_i_n inmove(sq,place);
                  sqo:=sq; sq:=looki:=b[sq,j];
                  _i_f gier(look)=1 _t_h_e_n
                  _b_e_g_i_n
                     _i_f state_<2 _t_h_e_n
                     _b_e_g_i_n stinc:=alt:=0; stp:=stacktop+2; state:=3_e_n_d;
                     inmove(sq,sqo);
                     move(sq,_t_r_u_e,bad,niveau+1);
                     _i_f maxniv<niveau _t_h_e_n maxniv:=niveau;
                     _i_f maxniv=niveau _t_h_e_n tostack _e_l_s_e maxniv:=maxniv-1;
                     reestab(2)
                  _e_n_d _e_l_s_e reestab(1)
               _e_n_d case 3

            _e_n_d case
         _e_n_d
      _e_n_d almindelig brik _e_l_s_e
      _b_e_g_i_n _c_o_m_m_e_n_t dam;

         _f_o_r j:=3 _s_t_e_p 1 _u_n_t_i_l 6 _d_o
         _i_f bad|=j _t_h_e_n
         _b_e_g_i_n sqo:=place; sq:=looki:=b[place,j];
            try:
            _c_a_s_e gier(look) _o_f
            _b_e_g_i_n

               _b_e_g_i_n _c_o_m_m_e_n_t frit felt;
                  _i_f -,take ∧ state_<2 _t_h_e_n
                  _b_e_g_i_n state:=2;
                     inmove(sq,sqo); tostack; reestab(1);
                     sqo:=sq;
                     _f_o_r looki:=b[sq,j] _w_h_i_l_e gier(look)=1 _d_o
                     _b_e_g_i_n sq:=looki; inmove(sq,sqo);tostack; reestab(1); sqo:=sq _e_n_d;
                     sq:=looki
                  _e_n_d _e_l_s_e
                  _b_e_g_i_n sqo:=sq;
                     _f_o_r looki:=b[sq,j] _w_h_i_l_e gier(look)=1 _d_o sqo:=sq:=looki;
                     sq:=looki
                  _e_n_d;
                  _g_o_t_o try
               _e_n_d case1;

               _c_o_m_m_e_n_t optaget af egen brik;;

               _b_e_g_i_n _c_o_m_m_e_n_t optaget af modstander;
                  inmove(sq,sqo);
                  sqo:=sq; sq:=looki:=b[sq,j];
                  _i_f gier(look)=1 _t_h_e_n
                  _b_e_g_i_n
                     _i_f state_<3 _t_h_e_n
                     _b_e_g_i_n stinc:=alt:=0;stp:=stacktop+2;state:=4_e_n_d;
                     inmove(sq,sqo);
                     move(sq,_t_r_u_e,(j-4)_m_o_d4+3,niveau+1);
                     _i_f maxniv<niveau _t_h_e_n maxniv:=niveau;
                     _i_f maxniv=niveau _t_h_e_n tostack _e_l_s_e maxniv:=maxniv-1;
                     reestab(2)
                  _e_n_d _e_l_s_ereestab(1)
               _e_n_d case 3

            _e_n_d case
         _e_n_d
      _e_n_d dam
   _e_n_d procedure move;

   _b_o_o_l_e_a_n _p_r_o_c_e_d_u_r_e comp;
   _c_o_m_m_e_n_t comp sammenligner et træk i stakken med hidtil bedste (hb)
   og har værdien true, hvis der er flere træk tilbage og false ellers.
   desuden reetablerer comp dambrættet;

   _b_e_g_i_n _i_n_t_e_g_e_r i,ii,j,k,c,niveau,p1,p2,ant;
      _b_o_o_l_e_a_n boo,boo1;
      _i_f _f_a_l_s_e _t_h_e_n _b_e_g_i_n writecr; writetext(|<<comp|>); _e_n_d;
      boo1:=_t_r_u_e;
      et mere:
      i:=_i_n_t_e_g_e_rstack[1];
      ant:=_i_n_t_e_g_e_rstack[i-1]-1;
      k:=i-ant-4;
      _f_o_r j:=3 _s_t_e_p 1 _u_n_t_i_l ant _d_o
      a[j]:=stack[k+j];
      niveau:=p1:=0; boo:=_t_r_u_e;
      _f_o_r i:=i _w_h_i_l_e i|=0 _d_o
      _b_e_g_i_n _c_o_m_m_e_n_t pointsammentælling ved linkning gennem stakken;
         p1:=p1+(_i_f boo _t_h_e_n _i_n_t_e_g_e_r stack[i-2] _e_l_s_e -_i_n_t_e_g_e_rstack[i-2]);
         ii:=i; i:=_i_n_t_e_g_e_rstack[i];
         boo:=-,boo;
         niveau:=niveau+1
      _e_n_d pointsammentælling_f_o_r;
      a[1]:=_b_o_o_l_e_a_nniveau;
      a[2]:=_b_o_o_l_e_a_np1;
      _i_f _i_n_t_e_g_e_rhb[1,niveau]=0 _t_h_e_n _g_o_t_o gem _e_l_s_e
      _b_e_g_i_n p2:=_i_n_t_e_g_e_rhb[2,niveau];
         _i_f(boo∧p1<p2)∨(-,boo∧p2<p1)∨(p1=p2∧random)_t_h_e_n
         gem:
         _b_e_g_i_n
            _f_o_r j:=1 _s_t_e_p 1 _u_n_t_i_l ant _d_o
            hb[j,niveau]:=a[j];hb[j,niveau]:=_f_a_l_s_e
         _e_n_d
      _e_n_d;

      vrkd:
      ant:=_i_n_t_e_g_e_rstack[ii-1];
      j:=ii-ant;
      k:=ii-4;
      _i_f boo1 _t_h_e_n move back(stack,j,k);
      c:=oldtop-1;
      boo1:=c+3=j;
      _i_f boo1∧oldtop=3 _t_h_e_n
      _b_e_g_i_n comp:=_t_r_u_e; _g_o_t_o exit _e_n_d;
      _i_f -, boo1 _t_h_e_n
      _b_e_g_i_n stack[c-1]:=_b_o_o_l_e_a_nj;
         stack[j]:=_f_a_l_s_e;
         stack[j+1]:=_b_o_o_l_e_a_nc;
         stack[c+2]:=_b_o_o_l_e_a_n(_i_n_t_e_g_e_r(stack[c+2])-ant);
         stacktop:=stacktop-ant;
         _i_f stack[stacktop-5] _t_h_e_n _g_o_t_o et mere
      _e_n_d flere træk tilbage paa dette niveau _e_l_s_e
      _b_e_g_i_n stack[c-1]:=_f_a_l_s_e;
         stacktop:=oldtop;
         oldtop:=_i_n_t_e_g_e_rstack[c]+1;
         c:=_i_n_t_e_g_e_rstack[oldtop];
         niveau:=c+1;
         _i_f_i_n_t_e_g_e_rhb[1,c]|=c _t_h_e_n _g_o_t_o gem1 _e_l_s_e
         _b_e_g_i_n boo:=(c_m_o_d2)=1;
            p1:=_i_n_t_e_g_e_rhb[2,c]; p2:=_i_n_t_e_g_e_rhb[2,niveau];
            _i_f(boo∧p1<p2)∨(-,boo∧p2<p1)∨(p1=p2∧random)_t_h_e_n
            gem1:
            _b_e_g_i_n i:=1;
               hb[1,c]:=_b_o_o_l_e_a_nc;
               _f_o_r i:=i+1 _w_h_i_l_e i<3 ∨_i_n_t_e_g_e_rhb[i,niveau]|=0 _d_o
               hb[i,c]:=hb[i,niveau];hb[i,c]:=_f_a_l_s_e
            _e_n_d;hb[1,niveau]:=_f_a_l_s_e
         _e_n_d;
         ii:=stacktop-2;
         _g_o_t_o vrkd
      _e_n_d ikke flere træk paa dette niveau;
      comp:=_f_a_l_s_e;
      ant:=_i_n_t_e_g_e_rstack[stacktop-3]-3;
      ii:= stacktop-5-ant;
      k:=ii+ant-1;

      move ahead(stack,ii,k);
      exit:
   _e_n_d comp;

   _b_o_o_l_e_a_n_p_r_o_c_e_d_u_r_e random;
   _c_o_d_e ran;
   2,44
   pma1,mlnre
   gma1,sca1
   arna1,ck1
   grp-1,hvr2
   e:9901
   _e;

   _i_n_t_e_g_e_r _p_r_o_c_e_d_u_r_e readkoord;
   _c_o_d_ei;
   2,44
   panre1t58ITA
   sy58,
   hsre
   hvre7LTA
   is-57
   bss,hvre3
   sr48D
   ck10,grre4
   ck-1IOA
   ck1,src42
   hvre3LT
   hsre
   hvre7LTA
   is-10
   bss,hvre3
   ck9IOB
   ck1,pmre4
   grre4,arnre5
   srre4X
   mlre5X
   arc42NOB
   ck-1,grp-1
   hvre3LOC

   hvre3NOC
   arnre1,ca60
   e3:srnc42,grp-1
   hsre
   e7:ck10,grc54
   hvre8
   e1:qq
   e5:8
   e4:qq
   e:lyns3,ca13
   srnc42ITA
   nc60,ca58
   grre1,hvre
   hvreLZ
   hrs1
   e8:_e;

   _p_r_o_c_e_d_u_r_e typeboard;
   _b_e_g_i_n _i_n_t_e_g_e_r bi,i,j;
      _b_o_o_l_e_a_n s1,s2,s3,boo;
      boo:=_f_a_l_s_e;
      s1:=_b_o_o_l_e_a_n |<<×××××|>;
      s2:=_b_o_o_l_e_a_n |<<|>;
      s3:=_b_o_o_l_e_a_n |<<_,_,_,_,_,|>;
      writecr;
      _f_o_r i:=32 _s_t_e_p -4 _u_n_t_i_l 2 _d_o
      _b_e_g_i_n
         writetext(_i_fboo_t_h_e_n_s_t_r_i_n_gs1_e_l_s_e_s_t_r_i_n_gs2);
         _f_o_r j:=0 _s_t_e_p 1 _u_n_t_i_l 3 _d_o
         _b_e_g_i_n
            bi:=_i_n_t_e_g_e_r(_b_o_o_l_e_a_nb[i-j,1]_s_h_i_f_t3)+1;
            writetext(_c_a_s_ebi_o_f(_s_t_r_i_n_gs3,|<< gal |>,|<< gal |>,|<< gal |>,
            |<<_,_,_,_,_,|>,|<<_,_,s_,_,|>,|<<_,_,_,_,_,|>,|<<_,_,h_,_,|>));
            _i_fj<3_t_h_e_nwritetext(_s_t_r_i_n_gs1)
         _e_n_d;
         _i_f boo _t_h_e_nwritecr;
         writetext(_s_t_r_i_n_gs1);
         _i_f-,boo_t_h_e_nwritecr;
         _f_o_r j:=0 _s_t_e_p 1 _u_n_t_i_l 3 _d_o
         _b_e_g_i_n

            bi:=_i_n_t_e_g_e_r(_b_o_o_l_e_a_nb[i-j,1]_s_h_i_f_t3)+1;
            writetext(_c_a_s_ebi_o_f(_s_t_r_i_n_gs3,|<< gal |>,|<< gal |>,|<< gal |>,
            |<<_,_,s_,_,|>,|<<_,sss_,|>,|<<_,_,h_,_,|>,|<<_,hhh_,|>));
            _i_fj<3_t_h_e_nwritetext(_s_t_r_i_n_gs1);
         _e_n_d;_i_f-,boo_t_h_e_nwritetext(_s_t_r_i_n_gs1);
         writeinteger(|<_,dd|>,9-(i_:4));writecr;
         _i_fboo_t_h_e_nwritetext(_s_t_r_i_n_gs1);
         _f_o_r j:=0 _s_t_e_p 1 _u_n_t_i_l 3 _d_o
         _b_e_g_i_n bi:=_i_n_t_e_g_e_r(_b_o_o_l_e_a_nb[i-j,1]_s_h_i_f_t3)+1;
            writetext(_c_a_s_ebi_o_f(_s_t_r_i_n_gs3,|<< gal |>,|<< gal |>,|<< gal |>,
            |<<_,_,_,_,_,|>,|<<_,_,s_,_,|>,|<<_,_,_,_,_,|>,|<<_,_,h_,_,|>));
            _i_fj<3_t_h_e_nwritetext(_s_t_r_i_n_gs1)
         _e_n_d;_i_f-,boo_t_h_e_nwritetext(_s_t_r_i_n_gs1);
         writecr;boo:=-,boo
      _e_n_d;writecr;
      writetext(|<<_,_,|>);
      _f_o_rj:=8_s_t_e_p-1_u_n_t_i_l1_d_o
      _b_e_g_i_nwritechar(48+j);writetext(|<<_,_,_,_,|>)_e_n_d;
      writecr;
   _e_n_d typeboard;

   _p_r_o_c_e_d_u_r_e printboard;
   _b_e_g_i_n _i_n_t_e_g_e_r bi,i,j,sel;
      _b_o_o_l_e_a_n s1,s2,s3,boo;
      sel:=select(8);
      boo:=_f_a_l_s_e; s3:=_410_610_3_00;
      s1:=_410_610_644_644_644_644_644;
      s2:=_410_610_610_610_610_610_610;
      writecr;writecr;writechar(30);
      writechar(9);writetext(|<<HVID|>);
      writecr;writecr;
      _f_o_r i:=32 _s_t_e_p -4 _u_n_t_i_l 2 _d_o
      _b_e_g_i_n writetext(_i_fboo_t_h_e_n_s_t_r_i_n_gs1_e_l_s_e_s_t_r_i_n_gs2);
         _f_o_r j:=0 _s_t_e_p 1 _u_n_t_i_l 3 _d_o
         _b_e_g_i_n bi:=_i_n_t_e_g_e_r(_b_o_o_l_e_a_nb[i-j,1]_s_h_i_f_t3)+1;
            writetext(_c_a_s_ebi_o_f(_s_t_r_i_n_gs3,|<< gal |>,|<< gal |>,|<< gal |>,
            |<<_,_,_,_,_,|>,|<<_,_,s_,_,|>,|<<_,_,_,_,_,|>,|<<_,_,h_,_,|>));
            _i_fj<3_t_h_e_nwritetext(_s_t_r_i_n_gs1)
         _e_n_d;
         _i_fboo_t_h_e_nwritecr;

         writetext(_s_t_r_i_n_gs1);_i_f-,boo_t_h_e_nwritecr;
         _f_o_r j:=0 _s_t_e_p 1 _u_n_t_i_l 3 _d_o
         _b_e_g_i_n bi:=_i_n_t_e_g_e_r(_b_o_o_l_e_a_nb[i-j,1]_s_h_i_f_t3)+1;
            writetext(_c_a_s_ebi_o_f(_s_t_r_i_n_gs3,|<< gal |>,|<< gal |>,|<< gal |>,
            |<<_,_,s_,_,|>,|<<_,sss_,|>,|<<_,_,h_,_,|>,|<<_,hhh_,|>));
            _i_fj<3_t_h_e_nwritetext(_s_t_r_i_n_gs1);
         _e_n_d;_i_f-,boo_t_h_e_nwritetext(_s_t_r_i_n_gs1);
         writeinteger(|<_,dd|>,9-(i_:4));writecr;
         _i_fboo_t_h_e_nwritetext(_s_t_r_i_n_gs1);
         _f_o_r j:=0 _s_t_e_p 1 _u_n_t_i_l 3 _d_o
         _b_e_g_i_n bi:=_i_n_t_e_g_e_r(_b_o_o_l_e_a_nb[i-j,1]_s_h_i_f_t3)+1;
            writetext(_c_a_s_ebi_o_f(_s_t_r_i_n_gs3,|<< gal |>,|<< gal |>,|<< gal |>,
            |<<_,_,_,_,_,|>,|<<_,_,s_,_,|>,|<<_,_,_,_,_,|>,|<<_,_,h_,_,|>));
            _i_fj<3_t_h_e_nwritetext(_s_t_r_i_n_gs1)
         _e_n_d;_i_f-,boo_t_h_e_nwritetext(_s_t_r_i_n_gs1);
         writecr;boo:=-,boo
      _e_n_d;writecr;
      writetext(|<<_,_,|>);
      _f_o_rj:=8_s_t_e_p-1_u_n_t_i_l1_d_o
      _b_e_g_i_nwritechar(48+j);writetext(|<<_,_,_,_,|>)_e_n_d;
      writecr;writecr;writechar(30);
      writechar(9);writetext(|<<sort|>);writechar(66);
      select(sel)
   _e_n_d printboard;

   _p_r_o_c_e_d_u_r_e move back(a,i1,i2);
   _v_a_l_u_ei1,i2;_i_n_t_e_g_e_ri1,i2;
   _b_o_o_l_e_a_n _a_r_r_a_y a;
   _b_e_g_i_n _b_o_o_l_e_a_n move;_i_n_t_e_g_e_ri,v,t;
      _f_o_r i:=i1 _s_t_e_p 1 _u_n_t_i_l i2 _d_o
      _b_e_g_i_n move:=a[i];
         _f_o_r i:=i _w_h_i_l_e _i_n_t_e_g_e_rmove|=0 _d_o
         _b_e_g_i_n t:=_i_n_t_e_g_e_r(move∧_3_30_7m);
            v:=_i_n_t_e_g_e_r((move_s_h_i_f_t-10)∧_3m_3_70);
            move:=(move_s_h_i_f_t-10)∧_1_00_3_0m;b[t,1]:=v
         _e_n_d
      _e_n_d_f_o_r
   _e_n_d move back;

   _p_r_o_c_e_d_u_r_e move ahead(a,i1,i2);

   _v_a_l_u_e i1,i2;_i_n_t_e_g_e_r i1,i2;
   _b_o_o_l_e_a_n _a_r_r_a_y a;
   _b_e_g_i_n _b_o_o_l_e_a_n move;_i_n_t_e_g_e_ri,v,t;
      v:=_i_n_t_e_g_e_r((a[i1]∧_3_00_3m_70)_s_h_i_f_t-10);
      _f_o_r i:=i1 _s_t_e_p 1 _u_n_t_i_l i2 _d_o
      _b_e_g_i_n
         move:=a[i];
         t:=_i_n_t_e_g_e_r(move∧_3_30_7m);
         move:=move∧_3_0m_1_00_s_h_i_f_t-10;
         _f_o_r i:=i _w_h_i_l_e _i_n_t_e_g_e_rmove|=0 _d_o
         _b_e_g_i_n b[t,1]:=0;
            t:=_i_n_t_e_g_e_r(move∧_3_30_7m);
            move:=move∧_3_0m_1_00_s_h_i_f_t-10
         _e_n_d_f_o_r;b[t,1]:=0
      _e_n_d;
      _i_f t>28∨t<5 _t_h_e_n v:=_i_n_t_e_g_e_r(_b_o_o_l_e_a_nv∧_20_11_3_30);
      _i_f t_<0∨t>32_t_h_e_n _b_e_g_i_n printstack;_g_o_t_o exit_e_n_d;
      b[t,1]:=v
   _e_n_d move ahead;

   _b_o_o_l_e_a_n _p_r_o_c_e_d_u_r_e deadpos;
   deadpos:=((state=2∧niveau_>goodnes)∨state=1)∨niveau>25;

   _b_o_o_l_e_a_n _p_r_o_c_e_d_u_r_e printstack;
   _b_e_g_i_n_i_n_t_e_g_e_ri,ii,iii,l,tabul,sel,k,j,t,højde,bogst;
      _b_o_o_l_e_a_n move;
      sel:=select(8); i:=ii:=1;writetext(|<<stackudskrift:|>);
      writecr;writecr;
      _f_o_r i:=i  _w_h_i_l_e i|=0 _d_o
      _b_e_g_i_n tabul:=2;
         k:=_i_f_i_n_t_e_g_e_rstack[i]=0_t_h_e_n1_e_l_s_e3;
         _f_o_r j:=0_s_t_e_p1_u_n_t_i_lk_d_o
         _b_e_g_i_nwriteinteger(|<  -ddd|>,_i_n_t_e_g_e_rstack[i+j]);writecr_e_n_d;
         i:=_i_n_t_e_g_e_rstack[i];  iii:=i;
         _i_f i=0 _t_h_e_n _g_o_t_o exit;
         _f_o_r i:=i _w_h_i_l_e i|=ii+4 _d_o
         _b_e_g_i_n k:=i-4;tabul:=tabul+3;
            writechar(30);writechar(tabul);
            i:=i-_i_n_t_e_g_e_rstack[i-1];

            _f_o_r j:=i _s_t_e_p 1 _u_n_t_i_l k _d_o
            _b_e_g_i_n move:=stack[j];
               _f_o_ri:=i _w_h_i_l_e _i_n_t_e_g_e_r move|=0 _d_o
               _b_e_g_i_n t:=_i_n_t_e_g_e_r(move∧_3_30_7m);
                  højde:=(t-1)_:4+1;
                  bogst:=48+(2×t-1)_m_o_d8 +(højde+1)_m_o_d2;
                  writechar(bogst);writechar(9-højde);writechar(27);
                  move:=move∧_3_0m_1_00_s_h_i_f_t-10
               _e_n_d
            _e_n_d;
            _f_o_r l:=1 _s_t_e_p 1 _u_n_t_i_l 3 _d_o
            _b_e_g_i_n
               writeinteger(|<  -ddd|>,_i_fl=1_t_h_e_n(_i_fstack[k+1]_t_h_e_n1_e_l_s_e0)_e_l_s_e_i_n_t_e_g_e_r0);
            _e_n_d;writecr
         _e_n_d;writecr;writecr;writecr;i:=ii:=iii
      _e_n_d;
      exit:
      writecr;writecr;writecr;
      writetext(|<<udskrift af sammenligningsstack:|>);writecr;
      writecr;writetext(|<<  niveau   point    træk|>);writecr;
      _f_o_rj:=1_s_t_e_p1_u_n_t_i_l39_d_o
      _i_f _i_n_t_e_g_e_rhb[1,j]=j_t_h_e_n
      _b_e_g_i_n write(|<    -ddd|>,_i_n_t_e_g_e_rhb[1,j],_i_n_t_e_g_e_rhb[2,j]);
         writetext(|<<_,_,_,_,|>);
         move:=hb[3,j];
         _f_o_r i:=4,i+1_w_h_i_l_e _i_n_t_e_g_e_r move|=0 _d_o
         _b_e_g_i_n_f_o_r i:=i _w_h_i_l_e _i_n_t_e_g_e_r move|=0 _d_o
            _b_e_g_i_n  t:=_i_n_t_e_g_e_r(move∧_3_30_7m);
               højde:=(t-1)_:4+1;
               bogst:=48+(2×t-1)_m_o_d8+(højde+1)_m_o_d2;
               writechar(bogst);writechar(9-højde);writechar(27);
               move:=move∧_3_0m_1_00_s_h_i_f_t-10
            _e_n_d;move:=hb[i,j]
         _e_n_d;writecr
      _e_n_d;writechar(66);
      select(sel)
   _e_n_d printstack;

   l1:_i_f where(|<<dambræt|>,db)|=0 _t_h_e_n

   _b_e_g_i_n
      select(16);
      _f_o_r i:=1 _s_t_e_p 1 _u_n_t_i_l 32 _d_o
      _f_o_r j:=1 _s_t_e_p 1 _u_n_t_i_l 6 _d_o b[i,j]:=0;
      _f_o_r i:=0 _s_t_e_p 1 _u_n_t_i_l 10 _d_o
      _f_o_r j:=1 _s_t_e_p 1 _u_n_t_i_l 40 _d_o hb[i,j]:=_f_a_l_s_e;
      hb[10,40]:=_4_010000001;
      _i_f reserve(|<<dambræt|>,1)|=0 _t_h_e_n
      _b_e_g_i_n
         writetext(|<<fejl i reserve|>);
         _g_o_t_o fejl;
      _e_n_d;
      _i_f where(|<<dambræt|>,db)|=0 _t_h_e_n
      _b_e_g_i_n
         writetext(|<<fejl i reserve|>);
         _g_o_t_o fejl;
      _e_n_d;
igen:
      i:=read integer;
      _i_f i>0 _t_h_e_n
      _b_e_g_i_n
         b[i,1]:=_i_n_t_e_g_e_r ((_b_o_o_l_e_a_n read integer)_s_h_i_f_t -3);
         _f_o_r j:=2 _s_t_e_p 1 _u_n_t_i_l 6 _d_o b[i,j]:=read integer;
         _g_o_t_o igen
      _e_n_d;
      put(hb,db,1);
      put(b,db,1);
   _e_n_d læs b;
   l5: _i_fget(hb,db,1)<0_t_h_e_n_g_o_t_ol5;
   hb[10,40]:=_b_o_o_l_e_a_n(_i_n_t_e_g_e_rhb[10,40]-1);
   ran:=_i_n_t_e_g_e_rhb[10,40];
   l6:_i_fput(hb,db,1)<0_t_h_e_n_g_o_t_ol6;
   l2:_i_f get(b,db,1)< 0 _t_h_e_n _g_o_t_o l2;
   select(16);writechar(66);
   writetext(|<<   inputkonventioner for gier - damspil:

   <farve>        ::=<sort>| <hvid>

   spilleren taster hvilken farve brikker han
   ønsker at spille med. hvid begynder altid.

   <styrke>       ::= <heltal>,

   det tal der tastes angiver det minimale an-
   tal træk, som gier ser frem. tal _<1 opfattes
   som 1, og tal _>25 opfattes som 25. med styrke
   lig 2 tager det ca. 15 sek. for gier at trække,
   og med styrke lig 3 tager det ca. 2 min.  nor-
   malt vil styrke 3 give passende modstand.

   <din tur:>     ::= <startpos> <mellempos> <slutpos><
   <startpos>     ::= <pos>,
   <mellempos>    ::=  | <pos>| <pos>,<mellempos>
   <slutpos>      ::= <pos>
   <pos>          ::= <bogstav><højde>
   <bogstav>      ::= a| b| c| d| e| f| g| h
   <højde>        ::= 1| 2| 3| 4| 5| 6| 7| 8

   notationen er sædvanlig skakbrætnotation. hvis
   man prøver at taste en position ind som ikke er
   korrekt, svares der med annul, og hele trækket
   er annuleret. desuden kan annulering ske efter
   sædvanlig help 3 konvention.

   det kan anbefales at udstyre sig med et dambræt naar man vil
   spille med gier, men har man ikke det, vil kaon bevirke udskrift
   af et dambræt paa linieskriveren, hver gang gier har trukket.|>);
   writechar(66);printboard;
   select(17);writecr;_f_o_ri:=1_s_t_e_p1_u_n_t_i_l40_d_ohb[1,i]:=_f_a_l_s_e;
   writetext(|<<GIER - DAMSPIL|>);writecr;

   l4:
   writecr;
   writetext(|<<farvevalg: |>);j:=lyn;
   _i_f j=18 _t_h_e_n
   _b_e_g_i_n writetext(|<<ort|>); valg:=_f_a_l_s_e _e_n_d _e_l_s_e _i_f j=56 _t_h_e_n
   _b_e_g_i_n writetext(|<<vid|>); valg:=_t_r_u_e _e_n_d _e_l_s_e _g_o_t_o l4;
   writecr;writetext(|<<styrke: |>);goodnes:=readinteger;
   _i_f-, valg _t_h_e_n _g_o_t_o os;
   start:

   color:=valg;
   stacktop:=3;
   stack[1] := _3_9 0 _11;
   oldtop:= stacktop;
   stack[2] := _f_a_l_s_e;
   niveau := 0;
   lookforward(color);
   _i_f state=1 _t_h_e_n
   _b_e_g_i_n writetext(|<<jeg vinder dette spil|>); _g_o_t_o exit_e_n_d;
   om igen:
   i:=1; dm:=_f_a_l_s_e; q:=0;
   writecr; writetext(|<<din tur  |>);
   _c_o_m_m_e_n_t indlæsning af modstanders træk;
   char:=0;
   _f_o_r k:=k _w_h_i_l_e char |=17 _d_o
   _b_e_g_i_n k:=readkoord;
      _i_f char<0∨k>32∨k_<0 _t_h_e_n
      _b_e_g_i_n writecr;
         _f_o_r i:= 29,49,37,37,20,35,62 _d_o
         writechar(i); _g_o_t_o om igen
      _e_n_d;
      _i_f q=40 _t_h_e_n
      _b_e_g_i_n q:=0;draw[i]:=dm;
         dm:= _f_a_l_s_e; i:=i+1
      _e_n_d;
      dm:=dm∨(((_b_o_o_l_e_a_nk)∨(_b_o_o_l_e_a_nb[k,1]_s_h_i_f_t10))_s_h_i_f_tq);
      q:=q+10;
   _e_n_d;
   draw[i]:=dm;
   draw[i+1]:= _f_a_l_s_e; _c_o_m_m_e_n_t  slut indlæsning af træk;
   _c_o_m_m_e_n_t kontrol af modstanders træk;
   mi:=i; ii:=stacktop-3;
   nyttræk:
   ant:=_i_n_t_e_g_e_rstack[ii]-3;
   ii:=ii-ant-3;
   _i_f mi|=ant _t_h_e_n
   next:
   _b_e_g_i_n _i_f ii=4 _t_h_e_n
      _b_e_g_i_n writecr;
         writetext(|<<ej lovligt træk - prøv igen|>);

         _g_o_t_o omigen
      _e_n_d;
      _g_o_t_o nyt træk
   _e_n_d;
   di:=draw[1];
   _f_o_r i:=1 _s_t_e_p 1 _u_n_t_i_l ant _d_o
   _i_f _i_n_t_e_g_e_r di|=_i_n_t_e_g_e_rstack[i+ii] _t_h_e_n _g_o_t_onext
   _e_l_s_e di:=draw[i+1];

   _c_o_m_m_e_n_t modstanderens træk trækkes;
   move ahead(draw,1,mi);

   _c_o_m_m_e_n_t vi begynder at se os om efter et modtræk;
   os:stacktop:=oldtop:=3;
   stack[1]:=_3_90_11;stack[2]:=_f_a_l_s_e;niveau:=0;
   color:=-,valg;
   ii:=look forward(color);
   _i_f state=1 _t_h_e_n
   _b_e_g_i_n writetext(|<<du vinder dette spil|>); _g_o_t_o exit _e_n_d;
   _i_f alt=1 _t_h_e_n
   _b_e_g_i_n _f_o_r i:=ii _s_t_e_p 1 _u_n_t_i_l stacktop-6 _d_o
      hb[i-ii+3,1]:=stack[i];
      hb[i-ii+3,1]:=_f_a_l_s_e
   _e_n_d _e_l_s_e
   _b_e_g_i_n
      move ahead(stack,ii,stacktop-6);
      slut:=_f_a_l_s_e;
      _f_o_r i:=i _w_h_i_l_e -,slut _d_o
      _b_e_g_i_n _f_o_r i:=i _w_h_i_l_e -,dead pos _d_o
         _b_e_g_i_n color:=(stack[oldtop]_s_h_i_f_t-1)_=valg;
            ii:=look forward(color);
            _i_f state>1 _t_h_e_n move ahead(stack,ii,stacktop-6);
         _e_n_d_f_o_r;_i_fkbon_t_h_e_nprintstack;
         slut:=comp; niveau:=_i_n_t_e_g_e_rstack[oldtop];
         state:=_i_fstack[stacktop-5]_t_h_e_n1_e_l_s_e3
      _e_n_d
   _e_n_d;
   _c_o_m_m_e_n_t paa dette sted i programmet skulle vi
   have fundet det bedste træk,som derfor lægges ind nu;
   writecr;writetext(|<<jeg trækker |>);

   hb[1,1]:=_f_a_l_s_e;
   i:=2; m:=hb[3,1];
   v:=(m_s_h_i_f_t-10)∧_3m_3_70;
   t:=_i_n_t_e_g_e_r(m∧_3_30_7m);
   højde:=((_i_n_t_e_g_e_rt)-1)_:4+1;
   bogst:=48+((2×(_i_n_t_e_g_e_rt))-1)_m_o_d8+(højde+1)_m_o_d2;
   writechar(bogst);writechar(9-højde);
   m:=(m_s_h_i_f_t-10)∧_1_00_3_0m;
   _f_o_ri:=i+1 _w_h_i_l_e _i_n_t_e_g_e_rm|=0 _d_o
   _b_e_g_i_n _f_o_r i:=i _w_h_i_l_e _i_n_t_e_g_e_rm|=0 _d_o
      _b_e_g_i_nb[t,1]:=0;
         t:=_i_n_t_e_g_e_r(m∧_3_30_7m);
         højde:=((_i_n_t_e_g_e_rt)-1)_:4+1;
         bogst:=48+((2×(_i_n_t_e_g_e_rt))-1)_m_o_d8+(højde+1)_m_o_d2;
         writechar(27);writechar(bogst);
         writechar(9-højde);
         m:=(m_s_h_i_f_t-10)∧_1_00_3_0m;
      _e_n_d;
      m:=hb[i+1,1]
   _e_n_d;writechar(17);
   _i_f t<5∨t>28 _t_h_e_n v:=v∨_20_11_3_30;
   _i_f t<1∨t>32 _t_h_e_n _b_e_g_i_n printstack; _g_o_t_o exit _e_n_d;
   b[t,1]:=_i_n_t_e_g_e_rv;
   _i_fkaon_t_h_e_nprintboard;
   typeboard;
   _g_o_t_o start;

   exit:
   writecr; writetext(|<<skal vi prøve igen|>);j:=lyn;
   _i_fj=33_t_h_e_n_b_e_g_i_nwritechar(49);writecr;_g_o_t_ol2_e_n_d_e_l_s_e_i_fj=37_t_h_e_n
   _b_e_g_i_nwritechar(53);writechar(33);writecr_e_n_d_e_l_s_e _g_o_t_o exit;

   writetext(|<<tak for spillet|>);writecr;
fejl:
_e_n_dt<
Ff
 1, 4, 1, 0, 0, 5, 0,
 2, 4, 2, 0, 0, 6, 5,
 3, 4, 2, 0, 0, 7, 6,
 4, 4, 2, 0, 0, 8, 7,

 5, 4, 4, 2, 1,10, 9,
 6, 4, 4, 3, 2,11,10,
 7, 4, 4, 4, 3,12,11,
 8, 4, 2, 4, 0,12, 0,
 9, 4, 2, 5, 0,13, 0,
10, 4, 4, 6, 5,14,13,
11, 4, 4, 7, 6,15,14,
12, 4, 4, 8, 7,16,15,
13, 0, 4,10, 9,18,17,
14, 0, 4,11,10,19,18,
15, 0, 4,12,11,20,19,
16, 0, 2,12, 0,20, 0,
17, 0, 2,13, 0,21, 0,
18, 0, 4,14,13,22,21,
19, 0, 4,15,14,23,22,
20, 0, 4,16,15,24,23,
21, 6, 4,18,17,26,25,
22, 6, 4,19,18,27,26,
23, 6, 4,20,19,28,27,
24, 6, 2,20, 0,28, 0,
25, 6, 2,21, 0,29, 0,
26, 6, 4,22,21,30,29,
27, 6, 4,23,22,31,30,
28, 6, 4,24,23,32,31,
29, 6, 2,26,25, 0, 0,
30, 6, 2,27,26, 0, 0,
31, 6, 2,28,27, 0, 0,
32, 6, 1,28, 0, 0, 0,
-1,